

Asymptotic Expansions

Devendra Kapadia
Wolfram Research, Inc.

Plan for the Talk

- History of Asymptotic Expansions
- Differential Equations
- Integrals
- Difference Equations
- Sums

History of Asymptotic Expansions

 17th century - exploration of calculus using infinite series

 18th century - pragmatic use of divergent series

 19th century - rigorous understanding of convergence

 Modern era - asymptotic expansions

Ordinary Points for Linear ODEs

Compute a Taylor polynomial approximation for Cos:

```
deqn = {y''[x] + y[x] == 0, y[0] == 1, y'[0] == 0};
sol = AsymptoticDSolveValue[deqn, y[x], {x, 0, 8}]
```

$$1 - \frac{x^2}{2} + \frac{x^4}{24} - \frac{x^6}{720} + \frac{x^8}{40320}$$

Plot the solution for various orders:



Regular Singular Points for Linear ODEs

Solve Bessel's equation of order 1/2 around the regular singular point $x = 0$:

```
besseleqn = x^2 y''[x] + x y'[x] + (x^2 - 1/4) y[x] == 0;
```

```
sol = AsymptoticDSolveValue[besseleqn, y[x], {x, 0, 6}]
```

$$\left(\sqrt{x} - \frac{x^{5/2}}{6} + \frac{x^{9/2}}{120}\right) \mathbb{c}_1 + \left(\frac{1}{\sqrt{x}} - \frac{x^{3/2}}{2} + \frac{x^{7/2}}{24}\right) \mathbb{c}_2$$

Plot the regular and singular solutions:



Irregular Singular Points for Linear ODEs

Solve the Airy equation at Infinity:

```
AsymptoticDSolveValue[y''[x] - x y[x] == 0, y[x], {x, ∞, 3}]
```

$$e^{-\frac{2x^{3/2}}{3}}\left(\frac{385}{4608\,x^{13/4}}-\frac{5}{48\,x^{7/4}}+\frac{1}{x^{1/4}}\right)\mathbb{c}_1+e^{\frac{2x^{3/2}}{3}}\left(\frac{385}{4608\,x^{13/4}}+\frac{5}{48\,x^{7/4}}+\frac{1}{x^{1/4}}\right)\mathbb{c}_2$$

Plot the approximations and the Airy functions:



Nonlinear Differential Equations

Compute a polynomial approximation for a nonlinear first-order ODE:

```
eqn = {3 y′[x]^2 + 4 x y′[x] - y[x] + x^2 == 0, y[0] == 1};
sol = Quiet[AsymptoticDSolveValue[eqn, y[x], {x, 0, 37}]]
```

$$1 - \frac{x}{\sqrt{3}} - \frac{x^2}{4}$$

Verify that this is an exact solution of the ODE:

```
Simplify[eqn /. {y → Function[{x}, Evaluate[sol]]}]
{True, True}
```

Boundary Layer Problems



Solve a singular perturbation problem:

```
eqn = {ϵ y''[x] + 2 y'[x] + y[x] == 0, y[0] == 0, y[1] == 1/2};
```

```
psol = AsymptoticDSolveValue[eqn, y[x], x, {ϵ, 0, 1}]
```

$$\frac{1}{2} e^{\frac{1-x}{2}} - \frac{1}{2} e^{\frac{1}{2}-\frac{2x}{\epsilon}}$$

Compare with the exact solution:

```
dsol = DSolveValue[eqn, y[x], x]
```

$$-\frac{e^{\frac{1}{\epsilon}+\frac{\sqrt{1-\epsilon}}{\epsilon}}\left(e^{x\left(-\frac{1}{\epsilon}-\frac{\sqrt{1-\epsilon}}{\epsilon}\right)}-e^{x\left(-\frac{1}{\epsilon}+\frac{\sqrt{1-\epsilon}}{\epsilon}\right)}\right)}{2\left(-1+e^{\frac{2\sqrt{1-\epsilon}}{\epsilon}}\right)}$$



Laplace Method for Integrals

Function involving an exponential kernel that has a maximum at x = 1:

```
f[x_] := ⅇ^(-ω (x^2-2 x)) (1 + x)^(5/2)
```



Leading term in the expansion of the integral using Laplace's method:

```
AsymptoticIntegrate[f[x], {x, 0, ∞}, {ω, ∞, 1}]
```

$$4\, e^{\omega} \sqrt{2\pi} \sqrt{\frac{1}{\omega}}$$

Compare with the numerical result:

```
% /. {ω → 30.}
```

1.95625×10^{13}

```
NIntegrate[Exp[-30 (x^2 - 2 x)] (1 + x)^(5/2), {x, 0, ∞}]
```

1.97153×10^{13}

Obtain a better approximation by computing an extra term:

```
AsymptoticIntegrate[f[x], {x, 0, ∞}, {ω, ∞, 2}]
```

$$\frac{15\, e^{\omega} \sqrt{\frac{\pi}{2}}}{8\, \omega^{3/2}} + \frac{4\, e^{\omega} \sqrt{2\pi}}{\sqrt{\omega}}$$

```
% /. {ω → 30.}
```

1.97153×10^{13}

Method of Stationary Phase

Leading term approximation for the integral of an oscillatory function:

```
f[x_] := ⅇ^(ⅈ ω Sin[t])
Plot[Re[f[x] /. {ω → 50}], {t, 0, π}, Filling → Axis, FillingStyle → Yellow]
```



The approximation is computed using the method of stationary phase:

```
int = AsymptoticIntegrate[f[t], {t, 0, π}, {ω, ∞, 1}]
```

$$\frac{e^{-\frac{i\pi}{4}+i\omega}\sqrt{2\pi}}{\sqrt{\omega}}$$

Compare with a numerical approximation:

```
int /. ω → 5000.
```

-0.0208877 - 0.0286416 ⅈ

```
NIntegrate[Exp[ⅈ 5000 Sin[t]],
  {t, 0, π}, MinRecursion → 20, MaxRecursion → 20]
```

-0.0208884 - 0.0282411 ⅈ

Series Expansions for Integrals

Consider the following series expansion:

```
aint = AsymptoticIntegrate[e^-t/(1 + x t), {t, 0, ∞}, {x, 0, 8}]
```

$1 - x + 2x^2 - 6x^3 + 24x^4 - 120x^5 + 720x^6 - 5040x^7 + 40320x^8$

The series diverges for all nonzero values of x:

```
a[n_] := (-1)^n n! x^n
Table[a[n], {n, 0, 8}]
```

$\{1, -x, 2x^2, -6x^3, 24x^4, -120x^5, 720x^6, -5040x^7, 40320x^8\}$

```
SumConvergence[a[n], n]
```

x == 0

A numerical comparison shows that the expansion is fairly accurate:

```
aint /. x → 0.05
```

0.954371

```
NIntegrate[e^-t/(1 + 0.05 t), {t, 0, ∞}]
```

0.954371

The approximation is no longer accurate with a larger number of terms:

```
AsymptoticIntegrate[e^-t/(1 + x t), {t, 0, ∞}, {x, 0, 150}] /. {x → 0.05`20}
```

$3.53245487755745350 \times 10^{67}$

Obtain the exact result using Integrate or using Regularization:

```
Integrate[e^-t/(1 + x t), {t, 0, ∞}, Assumptions → x > 0]
```

$-\frac{e^{\frac{1}{x}} \text{ExpIntegralEi}\left[-\frac{1}{x}\right]}{x}$

```
Sum[a[n], {n, 0, ∞}, Regularization → "Borel"]
```

$\frac{e^{\frac{1}{x}} \text{Gamma}\left[0, \frac{1}{x}\right]}{x}$

Exact Asymptotics!

Integrate returns unevaluated for this example:

$$\int_0^\infty \frac{1}{\sqrt{1+x^4}\left(1+x^{\text{GoldenRatio}}\right)}\,dx$$

$$\int_0^\infty \frac{1}{\sqrt{1+x^4}\left(1+x^{\text{GoldenRatio}}\right)}\,dx$$

Obtain an asymptotic expansion:

```
sol = AsymptoticIntegrate[1/(Sqrt[1 + x^4] (1 + x^α)), {x, 0, ∞}, {α, 0, 4}]
```

$$\frac{2\,\text{Gamma}\left[\frac{5}{4}\right]^2}{\sqrt{\pi}}$$

Verify that this is the exact result:

```
N[sol, 80]
```

0.92703733865068595921692517359763002310879941176088345279296402252801088884190599

```
NIntegrate[1/(Sqrt[1 + x^4] (1 + x^GoldenRatio)), {x, 0, ∞}, WorkingPrecision → 80]
```

0.92703733865068595921692517359763002310879941176088345279296402252801088884190599

Ordinary Points for Linear OΔEs

Taylor series solution for a linear OΔE with an ordinary point at Infinity:

```
AsymptoticRSolveValue[{y[n + 1] == ((n^3 + 1) y[n])/n^3}, y[n], {n, ∞, 2}]
```

$$\left(1 - \frac{1}{2 n^2}\right) c_1$$

Plot the solution:

```
DiscretePlot[% /. {c1 → 1/100}, {n, 1, 30}]
```



Regular Singular Points for Linear OΔEs

Frobenius series solution for a linear OΔE with a regular singular point:

```
AsymptoticRSolveValue[2 n y[n + 1] == (2 n + 1) y[n], y[n], {n, ∞, 3}]
```

$$\left(1 + \frac{5}{1024\, n^3} + \frac{1}{128\, n^2} - \frac{1}{8\, n}\right) \sqrt{n}\; c_1$$

Plot the solution:

```
DiscretePlot[% /. {c1 → 1}, {n, 1, 30}]
```



Irregular Singular Points for Linear OΔEs

Find an asymptotic approximation for the sequence of involutions:

```
peqn = a[n + 1] == a[n] + n a[n - 1];
```

Obtain the leading asymptotic term:

```
Clear[sol, a]
sol[n_] = AsymptoticRSolveValue[peqn, a[n], {n, ∞, 1}]
```

$$e^{-\sqrt{n}}\left(1-\frac{119}{1152\,n}-\frac{7}{24\sqrt{n}}\right)n^{n/2}\,c_2\,(-0.607\ldots)^n+e^{\sqrt{n}}\left(1-\frac{119}{1152\,n}+\frac{7}{24\sqrt{n}}\right)n^{n/2}\,c_1\,(0.607\ldots)^n$$

The second component of the expansion approaches 0 for large n:

$$\lim_{n\to\infty,\ n\in\mathbb{Z}} \text{sol}[n]$$

$$c_1\,\infty$$

Linear Systems of Difference Equations

Find a series solution for a linear system of ODEs at $n = \infty$:

```
In[]:= eqns = {y[n + 1] == w[n] + z[n], z[n + 1] == 2 y[n] - z[n], w[n + 1] == y[n]/(n + 1) + z[n]};

In[]:= sol = AsymptoticRSolveValue[eqns, {w[n], y[n], z[n]}, {n, ∞, 1}] //
         FullSimplify[#, n ∈ Integers] &
```

Out[]=

$$\left\{\frac{2^{-4+\frac{n}{2}}\left(\left(-3+12\sqrt{2}+16n\right)\mathbb{c}_1+(-1)^{1+n}\left(3+12\sqrt{2}-16n\right)\mathbb{c}_2\right)}{n^{3/4}}+(-1)^n\mathbb{c}_3,\right.$$
$$\frac{2^{-5+\frac{n}{2}}(1+16n)\left(\left(2+\sqrt{2}\right)\mathbb{c}_1+(-1)^{1+n}\left(-2+\sqrt{2}\right)\mathbb{c}_2\right)}{n^{3/4}},$$
$$\left.\frac{2^{\frac{1}{2}(-7+n)}\left(\left(-7+4\sqrt{2}+16n\right)\mathbb{c}_1+(-1)^n\left(7+4\sqrt{2}-16n\right)\mathbb{c}_2\right)}{n^{3/4}}+(-1)^{1+n}\mathbb{c}_3\right\}$$

Indefinite Sums

Compute an asymptotic expansion for an indefinite rational sum:

In[•]:= `AsymptoticSum[1/(k^2 + 1), k, {k, ∞, 8}]`

Out[•]= $\frac{1}{2 k^8} - \frac{23}{42 k^7} - \frac{1}{2 k^6} + \frac{1}{6 k^5} + \frac{1}{2 k^4} + \frac{1}{6 k^3} - \frac{1}{2 k^2} - \frac{1}{k}$

Compare with the result given by `Sum`:

In[•]:= `Sum[1/(k^2 + 1), k] // FullSimplify`

Out[•]= $-\frac{1}{2} i \left(\text{PolyGamma}[0, -i + k] - \text{PolyGamma}[0, i + k]\right)$

In[•]:= `Series[%, {k, ∞, 8}] // Normal`

Out[•]= $\frac{1}{2 k^8} - \frac{23}{42 k^7} - \frac{1}{2 k^6} + \frac{1}{6 k^5} + \frac{1}{2 k^4} + \frac{1}{6 k^3} - \frac{1}{2 k^2} - \frac{1}{k}$

Finite Sums

Compute an approximation for a finite q-rational sum:

```
In[•]:= AsymptoticSum[1/(2^k + 1), {k, 0, n}, {n, ∞, 1}]
```

Out[•]= $-\frac{1}{2\left(1+2^{1+n}\right)}+n-\frac{\text{Log}\left[1+2^{1+n}\right]}{\text{Log}[2]}+\frac{\text{QPolyGamma}\left[0,-\frac{i\pi}{\text{Log}[2]},\frac{1}{2}\right]}{\text{Log}[2]}$

Compute a numerical approximation for increasing values of n:

```
In[•]:= t1 = Table[%, {n, {5, 50, 500}}] // N[#, 20] & // Chop
Out[•]= {1.23443965962768200862, 1.26449978034844334646, 1.26449978034844420919}
```

Compare with the exact result:

```
In[•]:= Sum[1/(2^k + 1), {k, 0, n}]
```

Out[•]= $\frac{\text{QPolyGamma}\left[0,-\frac{i\pi}{\text{Log}[2]},\frac{1}{2}\right]}{\text{Log}[2]}-\frac{\text{QPolyGamma}\left[0,-\frac{i\pi-(1+n)\,\text{Log}[2]}{\text{Log}[2]},\frac{1}{2}\right]}{\text{Log}[2]}$

```
In[•]:= t2 = Table[%, {n, {5, 50, 500}}] // N[#, 20] & // Chop
Out[•]= {1.23357100415923945336, 1.26449978034844332101, 1.26449978034844420919}
```

```
In[•]:= t1 - t2
```

Out[•]= $\{0.0008686554684425553, 2.54\times10^{-17}, 0.\times10^{-20}\}$

Infinite Sums

Asymptotic expansion for an infinite harmonic sum:

In[∘]:= AsymptoticSum[$(-1)^k k e^{-k^2 x}$, {k, 1, ∞}, {x, 0, 3}]

Out[∘]= $-\frac{1}{4}-\frac{x}{8}-\frac{x^2}{8}-\frac{17 x^3}{96}$

Compare with a numerical approximation:

In[∘]:= % /. x → 0.07

Out[∘]= -0.259423

In[∘]:= NSum[$(-1)^k k e^{-k^2 0.07}$, {k, 1, ∞}]

Out[∘]= -0.259432

Final Remarks

- Asymptotics provides a powerful third mode of computation.

- We plan to implement a general Asymptotic wrapper soon.